[bookmark: _GoBack]Errata: Page 143 uses $clog twice in the 2nd paragraph. It should be $clog2 in both cases.


module tb_whileloop1();
   reg [7:0] tempreg;
   reg [3:0] count;
   integer loopcount;
   initial begin
    count = 0;
    tempreg = 8'bx;
    loopcount = 0;
    while (tempreg) begin
        #10 //delay to show running counts
        if (tempreg[0]) count = count + 1; 
        tempreg = tempreg >> 1; // Shift right
        loopcount = loopcount + 1;
        $display("tempreg = %b, count = %d, 
        loopcount = %d", tempreg, count, loopcount);
    end
  end
endmodule
Figure 5.1: Test fixture for determining true/false evaluation

module tb_whileloop1();
   reg [7:0] tempreg;
   reg [3:0] count;
   integer loopcount;
   initial begin
    count = 0;
    tempreg = 8'b1xxxxxxx;
    loopcount = 0;
    while (tempreg) begin
        #10 //delay to show running counts
        if (tempreg[0]) count = count + 1; 
        tempreg = tempreg >> 1; // Shift right
        loopcount = loopcount + 1;
        $display("tempreg = %b, count = %d, 
        loopcount = %d", tempreg, count, loopcount);
    end
  end
endmodule
Figure 5.2: Code of Figure 5.1 modified to set one bit of tempreg to logic one





parameter MAX_OUTER = 1023, MAX_INNER = 255;
integer i = 0, j = 0;
initial begin
  while (i <= MAX_OUTER) begin
	  while (j <= MAX_INNER) begin
		//run thousands of tests here
		j = j + 1;
    end
	i= i + 1;
  end
end
Figure 5.4: Typical usage of a while loop in a test fixture


module dowhile;
  initial begin
    int CNT = 0;
    do begin
      //CMINUS is local to the do...while loop
      int CMINUS;
      #1 CNT++;
      $strobe("COUNT = %d", CNT);
      CMINUS = CNT - 1;
      $strobe("CMINUS = %d", CMINUS);
    end
    while (CNT < 10);
  end
endmodule
Figure 5.5: Do while loop


module dowhile2;
  initial begin
    string CONDITION = "FALSE";
    do
      $display("This will print anyhow.");
      while (CONDITION != "FALSE");
   end
endmodule
Figure 5.6: Do while loops always run at least once.




module forloop(input [3:0] A, B, 
output reg [3:0] OUT);
  reg [2:0] I; 
  always @(A or B) begin
	 for (I = 0; I <= 3; I = I + 1)
		OUT[I] = A[I] & B[I];
  end
endmodule
Figure 5.7: A syntactically-correct but useless for loop


module forloop2 #(WIDTH = 8) (input CLOCK, RESET, LOAD, 
  input [WIDTH - 1:0] DATA, output reg [WIDTH - 1 : 0] COUNT);
reg [WIDTH - 1 : 0] INDEX, TEMP;
always @(posedge CLOCK or negedge RESET) begin
	if (!RESET) COUNT <= 0;
	else begin
	  if (LOAD) TEMP <= DATA;
		else begin
		  for (INDEX = 0; INDEX < WIDTH; INDEX = INDEX + 1) begin
		  	 COUNT <= COUNT + TEMP[0];
			  TEMP <= TEMP >> 1;
		  end
		 end
	end
end
endmodule
Figure 5.8: Counting ones with a for loop




module forloop2 #(WIDTH = 8) (input CLOCK, RESET, LOAD, 
  input [WIDTH - 1:0] DATA, output reg [WIDTH - 1 : 0] COUNT);
reg [WIDTH - 1 : 0] INDEX, TEMP;
always @(posedge CLOCK or negedge RESET) begin
	if (!RESET) COUNT <= 0;
	else begin
	  if (LOAD) TEMP <= DATA;
		else begin
		  for (INDEX = 0; |TEMP; INDEX = INDEX + 1) begin
		  	 COUNT <= COUNT + TEMP[0];
			  TEMP <= TEMP >> 1;
		  end
		 end
	end
end
endmodule
Figure 5.9: Non-synthesizable for loop.


module forloop2 #(WIDTH = 8) (input CLOCK, RESET, LOAD, 
  input [WIDTH - 1:0] DATA, output reg [WIDTH - 1 : 0] COUNT);
reg [WIDTH - 1 : 0] TEMP;
reg [$clog2(WIDTH) : 0] INDEX;
always @(posedge CLOCK or negedge RESET) begin
	if (!RESET) COUNT <= 0;
	else begin
	  if (LOAD) TEMP <= DATA;
		else begin
		  for (INDEX = 0; INDEX < WIDTH; INDEX = INDEX + 1) begin
		  	 COUNT <= COUNT + TEMP[0];
			  TEMP <= TEMP >> 1;
		  end
		 end
	end
end
endmodule
Figure 5.10: Sizing loop control variable with $clog2




/*Shift register using an unnecessary loop*/
module shiftreg #(DEPTH = 8) (input CLK, DATA_IN, 
  output DATA_OUT);
  reg [DEPTH - 1 : 0] SHIFT;
  assign DATA_OUT = SHIFT[DEPTH - 1];
  always_ff @(posedge CLK) begin
    SHIFT[0] <= DATA_IN;
    for (int I = 0; I < DEPTH; I++)
      SHIFT[I + 1] <= SHIFT[I];
  end
endmodule
  
/*Shift register without a loop. Both will
synthesize to the same hardware but this one
will simulate faster.*/
module SR2#(DEPTH = 8) (input CLK, DATA_IN, 
  output reg DATA_OUT);
  reg [DEPTH - 2 : 0] SHIFT;
  always_ff @(posedge CLK)
    {DATA_OUT, SHIFT} <= {SHIFT, DATA_IN};
endmodule
Figure 5.11: Shift registers with and without a loop


module foreachloop;
  reg [3:0] MEM [0:7];
  int i;
  initial begin
   $readmemh("memdata.txt", MEM);
    foreach(MEM[i])
      $display("MEM[%d] = %h", i, MEM[i]);
  end
endmodule
Figure 5.12: Foreach loop stepping through a memory array



module CLKGEN(output bit CLK);
  initial
    //using a bit variable, initialization to 0 is unnecessary
    forever #1 CLK = ~CLK;
endmodule
Figure 5. 13: Clock generator using a forever loop


module CLKGEN2(output bit CLK, CLK2);
  
  initial begin
    CLK = 1'b0; CLK2 = 1'b0;
    
      forever #1 CLK = ~CLK;
      //Error: CLK2 will never toggle
      forever #2 CLK2 = ~CLK2;
  end
endmodule
Figure 5.14: Defective dual clock generator



module CLKGEN2(output bit CLK, CLK2);
  
  initial begin
    CLK = 1'b0; CLK2 = 1'b0;
    fork
      forever #1 CLK = ~CLK;
      forever #2 CLK2 = ~CLK2;
    join
  end
endmodule
Figure 5.15: Two forever loops running in parallel


module repeatloop;
 initial begin
   int CNT = 4, LOOPCNT = 0;
     repeat (CNT) begin
     CNT++;
     LOOPCNT++;
     $display("CNT = %d, LOOPCNT = %d", CNT, LOOPCNT);
   end
 end
endmodule

/*Simulation Results:
# CNT = 5, LOOPCNT = 1
# CNT = 6, LOOPCNT = 2
# CNT = 7, LOOPCNT = 3
# CNT = 8, LOOPCNT = 4*/
Figure 5.16: A repeat loop






module foreachloop2;
  reg [3:0] MEM [0:7];
  int i;
  initial begin
   $readmemh("memdata.txt", MEM);
    foreach(MEM[i]) begin
      //skip printing on 4th & 6th words but stay in loop
      if (i == 3 || i == 5) continue;
      //quit the loop before printing the last element
      else if (i == 7) break;
      else $display("MEM[%d] = %h", i, MEM[i]);
    end
  end
endmodule
Figure 5.17: Using continue and break to modify loop execution



module genB2G #(SIZE = 8) (input [SIZE - 1 : 0] Bin, 
  output logic [SIZE - 1 : 0] Gray);
  
  always_comb Gray[SIZE - 1] = Bin[SIZE - 1];
  generate
    genvar I;
    for (I = 0; I < SIZE - 1; I++)
      begin: B2G_conv
        assign Gray[I] = Bin[I] ^ Bin[I+1];
    end: B2G_conv
   endgenerate
endmodule


module loopB2G #(SIZE = 8) (input [SIZE - 1 : 0] Bin, 
  output logic [SIZE - 1 : 0] Gray);
  
  always_comb begin
    Gray[SIZE - 1] = Bin[SIZE - 1];
    for (int I = 0; I < SIZE - 1; I++)
      Gray[I] = Bin[I] ^ Bin[I+1];
  end
endmodule
Figure 5.18: A generate loop and a for loop creating identical binary to Gray converters




module generateselect #(WIDTH = 8) 
(input [WIDTH - 1 : 0] A, B, 
output logic [WIDTH : 0] SUM);

generate
  if (WIDTH < 8)
    RIPPLE  #(WIDTH)  ADDER(A, B, SUM);
  else
    CLA     #(WIDTH)  ADDER(A, B, SUM);
endgenerate
endmodule
Figure 5.19: Selecting implementations based on operand size


module COUNT #(SIZE = 4) (input CLK, RST, EN,
  output logic [SIZE - 1 : 0] CNT);
  
  always_ff @(posedge CLK, negedge RST)
    if (!RST) CNT <= 0;
    else
      if (EN) CNT <= CNT + 1;
      else CNT <= CNT;
endmodule

module TOPCNT #(CHANNELS = 2, SIZE = 4)
  (input CLK, RST, input [CHANNELS - 1 : 0] EN,
  output logic [SIZE - 1 : 0] CNT [0 : CHANNELS - 1]);
  
  genvar I;
  generate
    for (I = 0; I < CHANNELS; I++) begin: CNTGEN
      COUNT #(.SIZE(SIZE)) C(CLK, RST, EN[I], CNT[I]);
    end
  endgenerate
endmodule
Figure 5.20: Using a generate loop to create a variable number of instances


module flipflop(input CLK, RST, D, output logic Q);
  always_ff @(posedge CLK, negedge RST)
    if (!RST) Q <= 0;
    else Q <= D;
endmodule
Figure 5.21: Asynchronously resettable flipflop inferred with “if” statement



module decoder1(input [7:0] DATA, output reg [2:0] CODE);
  always @(DATA)
		if (DATA == 8'b00000001) CODE = 0;
		else if (DATA == 8'b00000010) CODE = 1;
		else if (DATA == 8'b00000100) CODE = 2;
		else if (DATA == 8'b00001000) CODE = 3;
		else if (DATA == 8'b00010000) CODE = 4;
		else if (DATA == 8'b00100000) CODE = 5;
		else if (DATA == 8'b01000000) CODE = 6;
		else if (DATA == 8'b10000000) CODE = 7;
		else CODE = 3'bx;
endmodule
Figure 5.22: One-hot decoder using if…else if…else statements




module decoder2(input [7:0] DATA, 
  output reg [2:0] CODE, output reg ERROR);
  always @(DATA) 
	 if (DATA == 8'b00000001) begin
	   CODE = 3'd0;
	   ERROR = 1'b0;
	 end
	 else if (DATA == 8'b00000010) begin
	   CODE = 3'd1;
	   ERROR = 1'b0;
	 end
	 else if (DATA == 8'b00000100) begin
	   CODE = 3'd2;
	   ERROR = 1'b0;
	 end
	 else if (DATA == 8'b00001000) begin
	   CODE = 3'd3;
	   ERROR = 1'b0;
	 end
	 else if (DATA == 8'b00010000) begin
	   CODE = 3'd4;
	   ERROR = 1'b0;
	 end
	 else if (DATA == 8'b00100000) begin
	   CODE = 3'd5;
	   ERROR = 1'b0;
	 end
	 else if (DATA == 8'b01000000) begin
	   CODE = 3'd6;
	   ERROR = 1'b0;
	 end
	 else if (DATA == 8'b10000000) begin
	   CODE = 3'd7;
	   ERROR = 1'b0;
	 end
	 else begin
	   CODE = 3'bx;
	   ERROR = 1'b1;
	 end
endmodule
Figure 5.23: Multiple statements in “if…else” clauses must be surrounded by “begin…end”




module setinside(input [7:0] DATA, output reg HIT);
  always_comb
    if (DATA inside {0, 2, 4, [9:15], 8'b10000?0?}) HIT = 1'b1;
    else HIT = 1'b0;
endmodule 
Figure 5.24: Set membership conditional



module mux4_1(A, B, C, D, SEL, OUT);
  input A, B, C, D;
  input [1:0] SEL;
  output reg OUT;
  always_comb
    case (SEL)
      2'b00: OUT = A;
      2'b01: OUT = B;
      2'b10: OUT = C;
      default: OUT = D;
    endcase
endmodule
Figure 5.25: Four-to-one multiplexor using case


module decoder3(input [7:0] DATA, output reg [2:0] CODE);
  always_comb
	 case (DATA)
		8'b00000001: CODE = 0;
		8'b00000010: CODE = 1;
		8'b00000100: CODE = 2;
		8'b00001000: CODE = 3;
		8'b00010000: CODE = 4;
		8'b00100000: CODE = 5;
		8'b01000000: CODE = 6;
		8'b10000000: CODE = 7;
		default: CODE = 3'bx;
	endcase
endmodule
Figure 5.26: One-hot decoder using a case statement




module catcontrol(A, B, C, D, S1, S0, OUT);
  input A, B, C, D, S0, S1;
  output reg OUT;
  always_comb
    case ({S1, S0})
      2'b00: OUT = A;
      2'b01: OUT = B;
      2'b10: OUT = C;
      default: OUT = D;
    endcase
endmodule
Figure 5.27: Concatenating variables to control a case statement


//case 2’b0x covers both state == 00 and state == 01
always @(state)
  casex (state)
    2’b11: next_state = 2’b00;
    2’b10: next_state = 2’b11;
    2’b0x: next_state = 2’b01;
  endcase
Figure 5.28: Using a wild card to set a don’t care bit




module insidedemo(input [2:0] OPCODE,
  output logic [3:0] ENABLE_BUS);
  always_comb
    case (OPCODE) inside
      3'b0xx: ENABLE_BUS = 4'd0;
      3'b100: ENABLE_BUS = 4'd1;
      3'b101: ENABLE_BUS = 4'd2;
      default:  ENABLE_BUS = 4'd7;
    endcase
endmodule

module casexdemo(input [2:0] OPCODE,
  output logic [3:0] ENABLE_BUS);
  always_comb
    casex (OPCODE)
      3'b0xx: ENABLE_BUS = 4'd0;
      3'b100: ENABLE_BUS = 4'd1;
      3'b101: ENABLE_BUS = 4'd2;
      default:  ENABLE_BUS = 4'd7;
    endcase
endmodule

module tb_insidedemo;
  reg [2:0] OPCODE;
  logic [3:0] ENABLE_BUS1, ENABLE_BUS2;
  insidedemo  UUT1(OPCODE, ENABLE_BUS1);
  casexdemo  UUT2(OPCODE, ENABLE_BUS2);
  initial begin
    $monitor("OPCODE = %b, BUS1 = %d, BUS2 = %d", OPCODE, ENABLE_BUS1, ENABLE_BUS2);
    OPCODE = 3'bx;
    #1 OPCODE = 3'b0xx;
    #1 OPCODE = 3'b000;
    #1 OPCODE = 3'b001;
    #1 OPCODE = 3'b100;
  end
endmodule

/*Results: difference at time 0. Unknown opcode
takes the default clause with case inside but
matches the first branch for casex.
# OPCODE = xxx, BUS1 =  7, BUS2 =  0
# OPCODE = 0xx, BUS1 =  0, BUS2 =  0
# OPCODE = 000, BUS1 =  0, BUS2 =  0
# OPCODE = 001, BUS1 =  0, BUS2 =  0
# OPCODE = 100, BUS1 =  1, BUS2 =  1*/
Figure 5.29: Using case inside to avoid treating don’t know bits as don’t care


always_comb
  case (ALPHA)
  0: begin
    OUT1 = A;
    OUT2 = B;
  end
  1: begin
    OUT1 = C;
    OUT2 = D;
end
Figure 5.30: Multiple assignments need to be surrounded by begin…end


//This one DOES NOT WORK!!!

always_comb
  case (ALPHA)
    0: OUT1 = A;
    0: OUT2 = B;
    1: OUT1 = C;
    1: OUT2 = D;
  endcase
//Only one of each will happen
Figure 5.31:  Erroneous attempt to execute multiple assignments per case





module overlap1(CONTROL, ALPHA, BETA);
  input [3:0] CONTROL;
  output reg ALPHA, BETA;
  always_comb
    /*Overlapping clauses. Both enumerated clauses
    will be true for CONTROL = 1111. First will be
    executes, second will not.*/
    casez (CONTROL)
      4'b11??: begin
        ALPHA = 1;
        BETA = 0;
      end
      4'b??11: begin
        BETA = 1;
        ALPHA = 0;
      end
      default: begin
        ALPHA = 0;
        BETA = 0;
      end
    endcase
endmodule
Figure 5.32: Overlapping case branches



/*Latch D when CONTROL is high. Note that, unlike
a D flipflop, D must be in the sensitivity list
when using classic Verilog sensitivity lists.*/

module DLATCH(input D, CONTROL, output reg Q);
  always @CONTROL or D) 
    if (CONTROL) Q = D;
endmodule
Figure 5.33: D latch with an incompletely-specified if clause


/*Latch D when CONTROL is high, SystemVerilog style*/

module DLATCH(input D, CONTROL, output reg Q);
  always_latch 
    if (CONTROL) Q = D;
endmodule
Figure 5.34: D latch using an always_latch construct




module latchstate(STATE, NEXT_STATE);
  input [1:0] STATE;
  output reg [1:0] NEXT_STATE;
  
  localparam FETCH = 2'b00, DECODE = 2'b01, EXECUTE = 2'b10;
  
  /*This will create a two-bit latch. When STATE is 11,
  the previous value of NEXT_STATE will be held.*/
  always @(STATE)
  case (STATE)
    FETCH: NEXT_STATE = DECODE;
    DECODE: NEXT_STATE = EXECUTE;
    EXECUTE: NEXT_STATE = FETCH;
  endcase
endmodule
Figure 5.35: D latches with an incompletely-specified case statement


module nolatches(input [2:0] OPERATOR, 
  input [7:0] ICOUNT, PCOUNT, XCOUNT,
  output reg [7:0] DATA);
  always_comb begin
    DATA = 8'b0; //default assignment
    case (OPERATOR)
      3'b001: DATA = ICOUNT;
      3'b010: DATA = PCOUNT;
      3'b100: DATA = XCOUNT;
    endcase
  end
endmodule
Figure 5.36: Avoiding latches with an unconditional assignment prior to multiway branch


module nolatches2(input [2:0] OPERATOR,
  input [7:0] ICOUNT, PCOUNT, XCOUNT,
  output reg [7:0] DATA);
  always_comb begin
    DATA = 8'b0; //default assignment
    if (OPERATOR == 3'b001) DATA = ICOUNT;
    else if (OPERATOR == 3'b010) DATA = PCOUNT;
    else if (OPERATOR == 3'b100) DATA = XCOUNT; 
  end
endmodule  
Figure 5.37: Alternative method of avoiding latch generation




module uniquedemo (input [7:0] DATA, output logic [2:0] CODE);
  always_comb 
	 unique if (DATA == 8'b00000001) CODE = 0;
	 else if (DATA == 8'b00000010) CODE = 1;
	 else if (DATA == 8'b00000100) CODE = 2;
	 else if (DATA == 8'b00001000) CODE = 3;
	 else if (DATA == 8'b00010000) CODE = 4;
	 else if (DATA == 8'b00100000) CODE = 5;
	 else if (DATA == 8'b01000000) CODE = 6;
	 else CODE = 7;
endmodule
Figure 5.38: “Unique” instructs tools that there should be no overlapping conditions


module overlap2(CONTROL, ALPHA, BETA);
  input [3:0] CONTROL;
  output reg ALPHA, BETA;
  always_comb
    /*Clauses are overlapping but adding unique
    modifier instructs synthesizer not to make a
    priority encoder.*/
    unique casez (CONTROL)
      4'b11??: begin
        ALPHA = 1;
        BETA = 0;
      end
      4'b??11: begin
        BETA = 1;
        ALPHA = 0;
      end
      default: begin
        ALPHA = 0;
        BETA = 0;
      end
    endcase
endmodule
Figure 5.39: Using “unique” modifier to force parallel circuit synthesis




module priority_demo(input EN1, EN2, 
  output logic ALPHA, BETA);
  always_comb
    priority case ({EN2, EN1})
      2'b00: begin
        ALPHA = 0;
        BETA = 0;
      end
      2'b01: begin
        ALPHA = 1;
        BETA = 0;
      end
      2'b10: begin
        ALPHA = 0;
        BETA = 1;
      end
  endcase
endmodule
Figure 5.42: Latch avoidance with “priority” modifier




module priority_demo(input EN1, EN2, 
  output logic ALPHA, BETA);
  always_comb
    priority case ({EN2, EN1})
      2'b00: begin
        ALPHA = 0;
        BETA = 0;
      end
      /*What about BETA?
      A latch will be generated!
      The priority modifier prevents
      ALPHA from being latched but the
      missing value from this enumerated
      clause will cause a latch to be 
      generated for BETA.*/
      2'b01: begin
        ALPHA = 1;
      end
      2'b10: begin
        ALPHA = 0;
        BETA = 1;
      end
    endcase
endmodule
Figure 5.43: Priority modifier does not prevent all latches
