module STATEMAC(input CLK, RST, SERIAL, output logic UNLOCK);
  //STATES is an enumerated type
  typedef enum {A, B, C, D} STATES;
  //STATE and NEXT_STATE are two variables of type STATES
  STATES STATE, NEXT_STATE;
  
  /*Unlock when the sequence 1011 has been received.
  Reset to zero to ensure it does not initialize to 
  unlocked.*/
  always_ff @(posedge CLK, negedge RST)
    if (!RST) UNLOCK <= 1'b0;
    else UNLOCK <= (STATE == D && SERIAL);
      
    //State registers
    always_ff @(posedge CLK, negedge RST)
      if (!RST) STATE <= A;
      else STATE <= NEXT_STATE;
        
    //State logic
    always_comb
      case (STATE)
        A:  if (SERIAL) NEXT_STATE = B;
            else NEXT_STATE = A;
        B:  if (!SERIAL) NEXT_STATE = C;
            else NEXT_STATE = A;
        C:  if (SERIAL) NEXT_STATE = D;
            else NEXT_STATE = A;
        default: NEXT_STATE = A;
      endcase
endmodule
Figure 12.1: State machine




module tb_STATEMAC;
  bit CLK, RST, SERIAL;
  STATEMAC  UUT(CLK, RST, SERIAL, UNLOCK);
  initial forever #1 CLK = ~CLK;
  initial begin
    RST = 1'b1;
    #2 RST = 1'b0;
    #2 RST = 1'b1;
    #2 SERIAL = 1'b1;
    #2 SERIAL = 1'b0;
    #2 SERIAL = 1'b1;
    #4 SERIAL = 1'b0;
    #2 SERIAL = 1'b1;
    #2 SERIAL = 1'b0;
    #2 SERIAL = 1'b1;
    #2 SERIAL = 1'b0;
  end
endmodule
Figure 12.2: Test fixture for Figure 12.1




/*Direct architecture filter with data pipeline same depth
as order of filter. Output is same width as data input.*/

module direct #(WIDTH = 4, DEPTH = 4) 
  (input CLOCK, input signed [WIDTH - 1 : 0] DATA_IN, 
  output logic signed [WIDTH - 1 : 0] DATA_OUT);
  logic signed [WIDTH - 1 : 0] SR [DEPTH - 1 : 0];
  logic signed [2*WIDTH - 1 : 0] PRODUCTS [DEPTH - 1 : 0];
  parameter logic signed [WIDTH - 1 : 0] COE [DEPTH - 1 : 0] = {1,2,3,4};
  logic signed [2*WIDTH - 1 + $clog2(DEPTH) : 0] SUM;
     
  always_ff @(posedge CLOCK) begin
    SR[0] <= DATA_IN;
    SR[DEPTH - 1 : 1] <= SR[DEPTH - 2 : 0];
  end
    
  always_comb begin
    for (int I = 0; I < DEPTH; I++)
      PRODUCTS[I] = SR[I] * COE[I];
  end
  
  always_comb begin
    SUM = PRODUCTS[0];
    for (int I = 1; I < DEPTH; I++)
      SUM = SUM + PRODUCTS[I];
  end
  
 always_comb DATA_OUT = SUM[2*WIDTH - 1 + $clog2(DEPTH) -: WIDTH];
endmodule
Figure 12.4: Direct mode FIR filter




/*Direct format filter with data pipeline one less than 
filter order. Operates on input data. Output is full 
precision of internal calculation, no truncation.*/

module direct2 #(WIDTH = 4, DEPTH = 4) 
  (input CLOCK, input signed [WIDTH - 1 : 0] DATA_IN, 
  output logic signed [2*WIDTH - 1 + $clog2(DEPTH) : 0] DATA_OUT);
  
  logic signed [WIDTH - 1 : 0] SR [DEPTH - 2 : 0];
  logic signed [2*WIDTH - 1 : 0] PRODUCTS [DEPTH - 1 : 0];
  parameter logic signed [WIDTH - 1 : 0] COE [DEPTH - 1 : 0] = {1,2,3,4};
  logic signed [2*WIDTH - 1 + $clog2(DEPTH) : 0] SUM;
  
  always_ff @(posedge CLOCK) begin
    SR[0] <= DATA_IN;
    SR[DEPTH - 2 : 1] <= SR[DEPTH - 3 : 0];
  end
    
  always_comb begin
    PRODUCTS[0] = DATA_IN * COE[0];
    for (int I = 1; I < DEPTH; I++)
      PRODUCTS[I] = SR[I - 1] * COE[I];
    end
  
  always_comb begin
    SUM = PRODUCTS[0];
    for (int I = 1; I < DEPTH; I++)
      SUM = SUM + PRODUCTS[I];
  end
 
  always_comb DATA_OUT = SUM;  
 
endmodule
Figure 12.5: Another variation on the direct mode filter


/*Direct architecture filter with data pipeline one less
than order of filter. Operates on data input. Multipliers
are instantiated to allow hierarchical compilation.*/

module directH #(WIDTH = 4, DEPTH = 4) 
  (input CLOCK, input signed [WIDTH - 1 : 0] DATA_IN, 
  output logic signed [2 * WIDTH - 1 + $clog2(DEPTH) : 0] DATA_OUT);
  
  logic signed [WIDTH - 1 : 0] SR [DEPTH - 2 : 0];
  logic signed [2*WIDTH - 1 : 0] PRODUCTS [DEPTH - 1 : 0];
  parameter logic signed [WIDTH - 1 : 0] COE [DEPTH - 1 : 0] = {1,2,3,4};
  logic signed [2*WIDTH - 1 + $clog2(DEPTH) : 0] SUM;
     
  always_ff @(posedge CLOCK) begin
    SR[0] <= DATA_IN;
    SR[DEPTH - 2 : 1] <= SR[DEPTH - 3 : 0];
  end
    
  always_comb PRODUCTS[0] = DATA_IN * COE[0];
    
  generate
    genvar I;
    for (I = 1; I < DEPTH; I++)
      multiply  #(WIDTH) M(SR[I-1], COE[I], PRODUCTS[I]);      
  endgenerate
  
  always_comb begin
    SUM = PRODUCTS[0];
    for (int I = 1; I < DEPTH; I++)
      SUM = SUM + PRODUCTS[I];
  end
  
  always_comb DATA_OUT = SUM;     
endmodule
Figure 12.6: Direct mode hierarchical filter design


//Scalable multiplier

module multiply #(WIDTH = 4) (input signed [WIDTH - 1 : 0] OP1, OP2,
  output logic signed [2 * WIDTH - 1 : 0] PROD);
  
  always_comb PROD = OP1 * OP2;
endmodule
Figure 12.7: Multiplier module referenced in Figure 12.6




module transpose #(WIDTH = 4, DEPTH = 4) (input CLK, 
  input signed [WIDTH - 1 : 0] DATA_IN,
	output logic signed [2 * WIDTH - 1 + $clog2(DEPTH) : 0] DATA_OUT);
	parameter logic signed [WIDTH - 1 : 0] COE [0 : DEPTH - 1] = {1, 2, 3, 4};
	logic signed [2 * WIDTH - 1 : 0] PRODUCT [0 : DEPTH - 1];
	logic signed [2 * WIDTH - 1 + $clog2(DEPTH) : 0] PIPE[0 : DEPTH - 1];
	
	always_comb
	for (int I = 0; I < DEPTH; I++)
		PRODUCT[I] = DATA_IN * COE[I];
	
	always_ff @(posedge CLK) begin
		PIPE[0] <= PRODUCT[0];
		for (int I = 0; I < DEPTH; I++) PIPE[I + 1] <= PRODUCT[I + 1] + PIPE[I];
	end
	
  always_comb DATA_OUT = PIPE[DEPTH - 1];
endmodule
Figure 12.8: Transpose architecture FIR filter


/*Hierarchical transpose format filter. Uses adder and multiplier module instances
that can be compiled separately.

All computed sum of products bits are output, meaning the output is more than
twice as many bits as the input. An alternative output assignment statement is
commented out. If the commented out version is to be used, the output statement
also needs to be changed to WIDTH - 1 bits.*/



module transposeH #(WIDTH = 4, DEPTH = 4) (input CLOCK, 
  input signed [WIDTH - 1 : 0] DATA_IN,
	output logic signed [2 * WIDTH - 1 + $clog2(DEPTH) : 0] DATA_OUT);
	parameter logic signed [WIDTH - 1 : 0] COE [0 : DEPTH - 1] = {1, 2, 3, 4};
	logic signed [2 * WIDTH - 1 : 0] PRODUCT [0 : DEPTH - 1];
	logic signed [2 * WIDTH - 1 + $clog2(DEPTH) : 0] PIPE[0 : DEPTH - 1];
	
	always_ff @(posedge CLOCK)
		PIPE[0] <= PRODUCT[0];
	
	generate
		genvar I, J;
		for (I = 0; I < DEPTH; I++)	 begin
		  MULT  #(WIDTH) M(CLOCK, DATA_IN, COE[I], PRODUCT[I]);
		 end
		 for (J = 1; J < DEPTH; J++) begin
		  ADD  #(WIDTH, DEPTH) A(CLOCK, PRODUCT[J], PIPE[J-1], PIPE[J]);
		end
	endgenerate
	
  always_comb DATA_OUT = PIPE[DEPTH - 1];
endmodule
Figure 12.9: Hierarchical transpose format FIR filter


//Scalable multiplier with registered output
module MULT #(WIDTH = 4) (input CLOCK, input signed [WIDTH - 1 : 0] DATA_IN, COE,
  output logic signed [2 * WIDTH - 1 : 0] PRODUCT);
  
  always_ff @(posedge CLOCK) PRODUCT <= DATA_IN * COE;
  
endmodule

//Scalable adder with registered output
module ADD #(WIDTH = 4, DEPTH = 4) (input CLOCK, 
  input signed [2 * WIDTH - 1 : 0] OPERAND1,
  input signed [2 * WIDTH - 1 + $clog2(DEPTH) : 0] OPERAND2,
  output logic signed [2 * WIDTH - 1 + $clog2(DEPTH) : 0] SUM);
  
  always_ff @(posedge CLOCK) SUM <= OPERAND1 + OPERAND2;
  
endmodule
Figure 12.10: Adders and multipliers referenced in hierarchical transpose design


module tb_filter;
  parameter WIDTH = 4;
  parameter DEPTH = 4;
  reg CLK = 1'b0;
  reg signed [WIDTH - 1 : 0] DATA_IN;
  wire signed [2 * WIDTH - 1 + $clog2(DEPTH) : 0] DATA_OUT1,DATA_OUT2, 
  DATA_OUT3, DATA_OUT4, DATA_OUT5;
  
  direct #(WIDTH, DEPTH) UUT1(CLK, DATA_IN, DATA_OUT1);
  direct2 #(WIDTH, DEPTH) UUT2(CLK, DATA_IN, DATA_OUT2);
  directH #(WIDTH, DEPTH) UUT3(CLK, DATA_IN, DATA_OUT3);
  transpose #(WIDTH, DEPTH) UUT4(CLK, DATA_IN, DATA_OUT4);
  transposeH #(WIDTH, DEPTH) UUT5(CLK, DATA_IN, DATA_OUT5);
  
  initial forever CLK = #1 ~CLK;
  
  initial begin
    DATA_IN = 1;
    #2 DATA_IN = 2;
    #2 DATA_IN = 3;
    #2 DATA_IN = 4;
    #2 DATA_IN = 5;
    #2 DATA_IN = 6;
    #2 DATA_IN = 7;
  end
endmodule
Figure 12.11: Test fixture for direct and transpose filter architectures




module processor #(WIDTH = 4, DEPTH = 4) 
  (input CLOCK, VALID, input signed [WIDTH - 1 : 0] DATA_IN, 
  output logic signed [WIDTH - 1 : 0] DATA_OUT,
  output logic READY);
  
  parameter logic signed [WIDTH - 1 : 0] COE [DEPTH - 1 : 0] = {4,3,2,1};
  logic signed [2*WIDTH - 1 + $clog2(DEPTH) : 0] Accumulator;
  logic signed [WIDTH - 1 : 0] SR [DEPTH - 1 : 0];
  logic [$clog2(WIDTH) - 1 : 0] BITCOUNT;
  logic [$clog2(DEPTH) - 1 : 0] WORDCOUNT;
  logic START;
  
  always_ff @(posedge CLOCK)
    if (VALID) begin
      BITCOUNT <= 'b0;
      WORDCOUNT <= DEPTH - 1;
    end
    else begin
      if (BITCOUNT == WIDTH - 1) BITCOUNT <= 'b0;
      else BITCOUNT <= BITCOUNT + 1;
      if (BITCOUNT == WIDTH - 1 && WORDCOUNT > 0) WORDCOUNT <= WORDCOUNT - 1;
      else if (BITCOUNT == WIDTH - 1) WORDCOUNT <= WORDCOUNT - 1;
      else WORDCOUNT <= WORDCOUNT;
    end
      
  /*make shift register DEPTH deep to avoid having
  to operate on data input, which may not be stable
  for the entire processing period.*/
  always_ff @(posedge CLOCK) begin
    if (VALID) begin
      SR[0] <= DATA_IN;
      SR[DEPTH - 1 : 1] <= SR[DEPTH - 2 : 0];
    end
    else SR <= SR;
  end
  
  always_ff @(posedge CLOCK) begin
    START <= VALID;
    if (WORDCOUNT == 0 && BITCOUNT == WIDTH - 1) READY <= 1'b1;
    else READY <= 1'b0;
    if (START)
      Accumulator <= SR[DEPTH - 1] * COE[DEPTH - 1][0];
    else
      Accumulator <= Accumulator + ((SR[WORDCOUNT] << BITCOUNT) * COE[WORDCOUNT][BITCOUNT]);
  end
    
  /*Only output WIDTH most significant bits.*/        
  always_comb DATA_OUT = Accumulator[2*WIDTH - 1 + $clog2(DEPTH) -: WIDTH];     

endmodule
FIGURE 12.12 Processor style filter



module tb_proc;
  timeunit 1us;
  timeprecision 1ns;
  parameter PERIOD = 1.0;
  parameter WIDTH = 4;
  parameter DEPTH = 4;
  reg CLOCK = 1'b0;
  reg VALID = 1'b1;
  reg signed [WIDTH - 1 : 0] DATA_IN = 1;
  wire signed [WIDTH - 1 : 0] DATA_OUT;
  wire READY;
  reg [1:0] DATA2 = 0;
  int EXPECTED;
  
  processor  UUT(CLOCK, VALID, DATA_IN, DATA_OUT, READY);
    
  initial forever CLOCK = #(PERIOD/2.0) ~CLOCK;

  always @(posedge CLOCK) begin
    //respond to UUT with a new word as soon as old word has been processed
    VALID <= READY; 
    if (READY)  begin
      //send 1,2,3,4 repeating for data
      DATA2 <= DATA2 + 1; //will count 0 - 3
      DATA_IN <= DATA2 + 1; //will count 1 - 4
    end
  end
  
  always @(posedge READY) begin
    EXPECTED = UUT.SR[0] * UUT.COE[0] + 
      UUT.SR[1] * UUT.COE[1] + 
      UUT.SR[2] * UUT.COE[2] + 
      UUT.SR[3] * UUT.COE[3];
      $display("SR Contents are %d, %d, %d, %d", 
        UUT.SR[0], UUT.SR[1], UUT.SR[2], UUT.SR[3]);
      $display("Coefficients are %d, %d, %d, %d", 
        UUT.COE[0], UUT.COE[1], UUT.COE[2], UUT.COE[3]);
       
    if (UUT.Accumulator != EXPECTED)
      $display("Error: Expected %d, Received %d", EXPECTED,UUT.Accumulator);
    else
      $display("Huzzah!:  Expected %d, Received %d", EXPECTED,UUT.Accumulator);
      
  end
  
endmodule
FIGURE 12.13 Test fixture for processor style filter

/*Processor style filter with pipelined multiply-accumulate.
READY signals to sending side that this filter can accept a
new word. STOP signals that the filter output is complete for
a given word.*/

module processorM #(WIDTH = 4, DEPTH = 4) 
  (input CLOCK, VALID, input signed [WIDTH - 1 : 0] DATA_IN, 
  output logic signed [WIDTH - 1 : 0] DATA_OUT,
  output logic READY, STOP);
  
  parameter logic signed [WIDTH - 1 : 0] COE [DEPTH - 1 : 0] = {4,3,2,1};
  logic signed [2*WIDTH - 1 + $clog2(DEPTH) : 0] Accumulator;
  logic signed [2*WIDTH - 1 : 0] PRODUCT;
  logic signed [WIDTH - 1 : 0] SR [DEPTH - 1 : 0];
  logic [$clog2(DEPTH) - 1 : 0] COUNT;
  logic START, NEWPROD;
       
  /*make shift register DEPTH deep to avoid having
  to operate on data input, which may not be stable
  for the entire processing period.*/
  always_ff @(posedge CLOCK) begin
    if (VALID) begin
      SR[0] <= DATA_IN;
      SR[DEPTH - 1 : 1] <= SR[DEPTH - 2 : 0];
    end
    else SR <= SR;
  end
  
  always_ff @(posedge CLOCK) PRODUCT <= SR[COUNT] * COE[COUNT];
  
  always_ff @(posedge CLOCK) begin
    START <= VALID;
    NEWPROD <= START;
    STOP <= READY;
    if (VALID) COUNT <= 0;
    else COUNT <= COUNT + 1;
    if (NEWPROD) Accumulator <= PRODUCT;
    else if (!STOP) Accumulator <= Accumulator + PRODUCT;
    else Accumulator <= Accumulator;
    //READY needs to go high on WORDCOUNT == DEPTH - 1 for max. performance
    if (COUNT == DEPTH - 1) READY <= 1'b1;
    else READY <= 1'b0;
  end
  
  /*Only output WIDTH most significant bits.*/        
  always_comb DATA_OUT = Accumulator[2*WIDTH - 1 + $clog2(DEPTH) -: WIDTH];     

endmodule
FIGURE 12.14 Processor style filter with hardware multiplier


/*Test bench for processor style filter with hardware multiplier*/
module tb_processor_M;
  timeunit 1us;
  timeprecision 1ns;
  parameter PERIOD = 1.0;
  parameter WIDTH = 4;
  parameter DEPTH = 4;
  reg CLOCK = 1'b0;
  reg VALID = 1'b1;
  reg signed [WIDTH - 1 : 0] DATA_IN = 1;
  wire signed [WIDTH - 1 : 0] DATA_OUT;
  wire READY, STOP;
  reg [1:0] DATA2 = 0;
  int EXPECTED;
  
  processorM  UUT(.*);
    
  initial forever CLOCK = #(PERIOD/2.0) ~CLOCK;

  always @(posedge CLOCK) begin
    //respond to UUT with a new word as soon as old word has been processed
    VALID <= READY; 
    if (READY)  begin
      //send 1,2,3,4 repeating for data
      DATA2 <= DATA2 + 1; //will count 0 - 3
      DATA_IN <= DATA2 + 1; //will count 1 - 4
    end
  end
  
  always @(posedge STOP) begin
    EXPECTED = UUT.SR[0] * UUT.COE[0] + 
      UUT.SR[1] * UUT.COE[1] + 
      UUT.SR[2] * UUT.COE[2] + 
      UUT.SR[3] * UUT.COE[3];
      $display("SR Contents are %d, %d, %d, %d", 
        UUT.SR[0], UUT.SR[1], UUT.SR[2], UUT.SR[3]);
      $display("Coefficients are %d, %d, %d, %d", 
        UUT.COE[0], UUT.COE[1], UUT.COE[2], UUT.COE[3]);
       
    if (UUT.Accumulator != EXPECTED)
      $display("Error: Expected %d, Received %d", EXPECTED,UUT.Accumulator);
    else
      $display("Huzzah!:  Expected %d, Received %d", EXPECTED,UUT.Accumulator);
      
  end
  
endmodule
FIGURE 12.15 Test fixture for hardware multiplier processor filter

#!/usr/bin/perl
use warnings;
use strict;

if($#ARGV != 1) {
    print STDERR "You must specify width and depth.\n";
        exit 4;
}

my $filename = "COE" . $ARGV[0] . "x" . $ARGV[1] . ".txt";
    
open (OUTFILE, ">$filename") or die "cannot open output file \n";

print OUTFILE "{";
for (my $I = 0; $I < $ARGV[1]; ++$I) {
	print OUTFILE $ARGV[0];
	print OUTFILE "'d";
	print OUTFILE int rand(2**$ARGV[0]);
	if ($I < $ARGV[1] - 1) {print OUTFILE ","};
	print OUTFILE  "\n";
	}
print OUTFILE "};";

close (OUTFILE);
FIGURE 12.16 PERL script for pseudo-random coefficient generation


module randnums;
  int ALPHA, BETA, GAMMA, DELTA;
  initial begin
    //pseudo-random 32 bit number
    ALPHA = $random;
    //pseudo-random 32 bit number with a seed 
    BETA = $random(2);
    //pseudo-random number ranging from -9 to +9
    GAMMA = $random % 10;
    //pseudo-random number ranging from 0 to 9
    DELTA = {$random} % 10;
    $display ("ALPHA = %d, BETA = %d, GAMMA = %d, DELTA = %d", ALPHA, BETA, GAMMA, DELTA);
  end
 endmodule
FIGURE 12.17 Using $random to generate pseudo-random numbers


module svrands;
  int ALPHA, BETA;
  initial begin
    //set seed to 10 for deterministic sequence
    $srandom(10);
    //pseudo-random 32 bit unsigned number
    ALPHA = $urandom;
    //pseudo-random number from 2 to 25
    BETA = $urandom_range(2,25);
    $display ("ALPHA = %d, BETA = %d", ALPHA, BETA);
end  
endmodule
FIGURE 12.18 SystemVerilog pseudo-random generator functions


/*Scalable, synthesizable FIFO design*/

module fifo_top #(WIDTH = 4, DEPTH = 8)
	(input WR_CLK, RD_CLK, RST, WR_EN, RD_EN, 
	input [WIDTH - 1 : 0] DATA_IN,
	output [WIDTH - 1 : 0] DATA_OUT, output FULL, EMPTY);
	
	wire [$clog2(DEPTH) : 0] WR_BIN, RD_BIN;
	wire WR_AASD, RD_AASD;
	
	//Synthesizable memory array
	regfile	MEM(WR_CLK, (WR_EN && !FULL), DATA_IN, WR_BIN[$clog2(DEPTH) - 1: 0], 
		RD_BIN[$clog2(DEPTH) - 1: 0], DATA_OUT);
	
	//Write side reset synchronizer
	aasd	WR_AA(WR_CLK, RST, WR_AASD);
	//Read side reset synchronizer
	aasd	RD_AA(RD_CLK, RST, RD_AASD);
	//Write pointer counter
	counter	WR_PTR(WR_CLK, WR_AASD, (WR_EN && !FULL), WR_BIN);
	//Read pointer counter
	counter	RD_PTR(RD_CLK, RD_AASD, (RD_EN && !EMPTY), RD_BIN);
	//Hierarchical flag generator subsysem
	flags	FLAGS(WR_CLK, RD_CLK, WR_AASD, RD_AASD,
			WR_BIN, RD_BIN, FULL, EMPTY);
endmodule
FIGURE 12.19 Top level of hierarchical FIFO design


/*Data storage register file
One read port, one write port*/

module regfile #(WIDTH = 4, DEPTH = 8) 
	(input WR_CLK, WR_EN, input [WIDTH - 1 : 0] DATA_IN, 
	input [$clog2(DEPTH) - 1 : 0] WR_ADDR, RD_ADDR,
	output logic [WIDTH - 1 : 0] DATA_OUT);
	
	reg	[WIDTH - 1 : 0] MEM [0 : DEPTH - 1];
	
	always_ff @(posedge WR_CLK) 
		if (WR_EN) MEM[WR_ADDR] <= DATA_IN;
		else MEM[WR_ADDR] <= MEM[WR_ADDR];
	always_comb DATA_OUT = MEM[RD_ADDR];
endmodule
FIGURE 12.20 FIFO register file


//Two stage reset synchronizer
module aasd(input CLK, RST, output logic SYNC_RST);
	logic STAGE1;
	always_ff @(posedge CLK, negedge RST)
		if (!RST) begin
			SYNC_RST <= 1'b0;
			STAGE1 <= 1'b0;
		end
		else begin
			SYNC_RST <= STAGE1;
			STAGE1 <= 1'b1;
		end
endmodule
FIGURE 12.21 Reset synchronizer





/*Generic, scalable binary counter.
It will be the log2 ceiling of parameter DEPTH bits.*/

module counter	#(DEPTH = 8) (input CLK, RST, EN, output logic [$clog2(DEPTH) : 0]CNT);
	always_ff @(posedge CLK, negedge RST)
	if (!RST) CNT <= '0;
	else
		if (EN) CNT <= CNT + 1;
		else CNT <= CNT;
endmodule
FIGURE 12.22 Generic counter used for address pointers
/*Hierarchical model of empty and full flag logic.
Full flag is write side, so uses write clock.
Empty flag is read side, so uses read clock.
Both sides use a two stage synchronizer on their
Gray bus inputs.
A Binary to Gray converter function is included to
convert the Gray code that crosses the clock domain
boundaries to binary for easy comparison of unequal
values.*/

module flags #(DEPTH = 8) (input WR_CLK, RD_CLK, WR_AASD, RD_AASD,
			input [$clog2(DEPTH) : 0] WR_BIN, RD_BIN,
			output FULL, EMPTY);
			wire [$clog2(DEPTH) : 0] RD_GRAY_SYNC, WR_GRAY_SYNC;
			
			sync	WR_SYNC(WR_CLK, B2G(RD_BIN), RD_GRAY_SYNC);
			sync	RD_SYNC(RD_CLK, B2G(WR_BIN), WR_GRAY_SYNC);
	
			full_logic	FL(WR_CLK, WR_AASD, WR_BIN, RD_GRAY_SYNC, FULL);
			empty_logic	EM(RD_CLK, RD_AASD, B2G(RD_BIN), WR_GRAY_SYNC, EMPTY);
			 
			function [$clog2(DEPTH) : 0] B2G(input [$clog2(DEPTH) : 0] BIN);
			B2G[$clog2(DEPTH) ] = BIN[$clog2(DEPTH) ];
			for (int I = $clog2(DEPTH) - 1; I >= 0; I--)
				B2G[I] = BIN[I] ^ BIN[I + 1];
			endfunction
endmodule
FIGURE 12.23 Hierarchical subdesign for flag logic




/*Scalable full flag design. Uses Gray to binary converts.*/

module full_logic #(DEPTH = 8) (input CLK, RST, 
	input [$clog2(DEPTH) : 0] WR_BIN, RD_GRAY, 
	output reg FULL);
	logic [$clog2(DEPTH) : 0] RD_BIN;
	
	function [$clog2(DEPTH) : 0] G2B(input [$clog2(DEPTH) : 0] GRAY);
				G2B[$clog2(DEPTH) ] = GRAY[$clog2(DEPTH) ];
				G2B[$clog2(DEPTH) - 1] =  GRAY[$clog2(DEPTH) ] ^  GRAY[$clog2(DEPTH) - 1];
				for (int I = $clog2(DEPTH) - 2; I >= 0; I--)
					G2B[I] = G2B[I+1] ^ GRAY[I];
	endfunction
	
	always_comb RD_BIN = G2B(RD_GRAY);
	
	always_comb
			if (WR_BIN[$clog2(DEPTH)] != RD_BIN[$clog2(DEPTH)] &&
				WR_BIN[$clog2(DEPTH) - 1 : 0] == RD_BIN[$clog2(DEPTH) - 1 : 0])
				FULL <= 1'b1;
			else
				FULL <= 1'b0;
endmodule
FIGURE 12.24 Full flag generator


/*Scalable comparator. Returns true (logic one)
when two inputs are equal.*/

module empty_logic #(DEPTH = 8) (input CLK, RST, 
	input [$clog2(DEPTH): 0] RD_GRAY, WR_GRAY, 
	output reg EMPTY);
	
	always_comb EMPTY <= RD_GRAY == WR_GRAY;
	
	always_comb CHECK_FLAG: assert (RD_GRAY == WR_GRAY && !EMPTY) 
	 $display("Error: at time %d empty flag should be set but is not", $time);
		
endmodule
FIGURE 12.25 Empty flag generator




/*Scalable two stage synchronizer.*/

module sync #(DEPTH = 8)(input CLK, input [$clog2(DEPTH) : 0] DATA, 
	output logic [$clog2(DEPTH) : 0] SYNC_DATA);
	
	logic [$clog2(DEPTH) : 0] STAGE1;
	always_ff @(posedge CLK) begin
		STAGE1 <= DATA;
		SYNC_DATA <= STAGE1;
	end
endmodule
FIGURE 12.26 Scalable synchronizer



module tb_fifo;
	timeunit 1ns;
	parameter WIDTH = 4;
	reg RD_CLK, WR_CLK, RST;
	reg WR_EN, RD_EN;
	reg [WIDTH - 1 : 0] DATA_IN;
	wire FULL, EMPTY;
	wire [WIDTH - 1 : 0] DATA_OUT;
	
	fifo_top	UUT(WR_CLK, RD_CLK, RST, WR_EN, RD_EN, DATA_IN, DATA_OUT, FULL, EMPTY);
	
	initial $monitor("At %d, RD_BIN = %b, RD_GRAY = %b, WR_BIN = %b, WR_GRAY = %b", 
	 $time, UUT.RD_BIN, UUT.FLAGS.RD_GRAY_SYNC, UUT.WR_BIN, UUT.FLAGS.WR_GRAY_SYNC);
		
	initial begin
		RD_CLK = 1'b1;
		forever #4 RD_CLK = ~RD_CLK;
	end
	
	initial begin
		WR_CLK = 1'b1;
		forever #1 WR_CLK = ~WR_CLK;
	end
	
	initial begin
		RST = 1'b1;
		# 3 RST = 1'b0;
		# 12 RST = 1'b1;
	end
	
	initial begin
		RD_EN = 1'b0; WR_EN = 1'b0;
		#20 WR_EN = 1'b1; RD_EN = 1'b1;
	end
	
	always_ff @(posedge WR_CLK, negedge RST)
	if (!RST) DATA_IN <= 0;
	else if (WR_EN && !FULL) DATA_IN <= DATA_IN + 1;
	else DATA_IN <= DATA_IN;
	
endmodule
FIGURE 12.27 FIFO test fixture


/*Top level of the DMX512 receiver.*/

module dmx(input CLK, RST, DATA_IN, input [8:0] SLOT, 
  output wire [7:0] PACKET, output wire VALID);
  wire SYNC_PULSE, BIT_EN, FRAME_EN;
  wire [3:0] BITCNT;
  wire [8:0] FRAME;
  sync    sync(CLK, DATA_IN, DATA2);
  detect  d0(CLK, DATA2, SYNC_DATA, SYNC_PULSE);
  start_det sd(CLK, RST, BIT_EN, SYNC_DATA, START);
  hex_cnt c0(CLK, RST, SYNC_PULSE, BIT_EN);
  bit_cnt c1(CLK, RST, BIT_EN, START, SYNC_DATA, BITCNT, FRAME_EN);
  frame_cnt c2(CLK, RST, START, (FRAME_EN && BIT_EN && !SYNC_DATA), FRAME);
  data_reg  r0(CLK, RST, SYNC_DATA, BIT_EN, FRAME_EN, BITCNT, FRAME, SLOT, PACKET, VALID);
endmodule
FIGURE 12.28 Top-level DMX512 receiver


/*Two stage synchronizer.
The output will be synchronized to the receiving clock. This
type of synchronizer will ensure to a high probability
that indeterminate, metastable values will not propagate
but the output is not guaranteed to be any specific value.*/

module sync(input CLK, DIN, output reg DOUT);
  reg STAGE1;
  always_ff @(posedge CLK) begin
    STAGE1 <= DIN;
    DOUT <= STAGE1;
  end
endmodule
FIGURE 12.29 Two-stage synchronizer for serial data




/*Start sequence detector.
A valid start sequence is at least 22 bits of logic zero
followed by at least 2 bits of logic one.
Once that sequence has been met, the next logic zero 
will be the start bit of the first slot.
*/

/*Unusual case: suppose a start sequence is immediately followed
by another start sequence. In that case, the first logic zero
should cause ZCNT to go to one. Need to cover this potential event.
*/

module start_det(input CLK, RST, EN, DATA_IN, output logic START);
  reg [4:0] ZCNT; //Zeros count
  reg [1:0] OCNT; //Ones count
  always_ff @(posedge CLK, negedge RST)
    if (!RST) ZCNT <= 0;
    else //increment if DATA_IN == 0 and not yet at 22
      if (EN && START && !DATA_IN) ZCNT <= 0;
      else if (EN && DATA_IN && ZCNT < 22) ZCNT <= 0;
      else if (EN && ZCNT == 22) ZCNT <= 22;
      else if (EN) ZCNT <= ZCNT + 1;
      else ZCNT <= ZCNT;
      
  always_ff @(posedge CLK, negedge RST)
    if (!RST) OCNT <= 0;
    else
      if (EN && !DATA_IN) OCNT <= 0;
      else if (EN && OCNT == 2 && DATA_IN) OCNT <= 2;
      else if (EN && ZCNT == 22 && DATA_IN) OCNT <= OCNT + 1;
      else OCNT <= OCNT;
   
   always_comb START = (ZCNT == 22 && OCNT == 2);   
endmodule
FIGURE 12.30 Start sequence detector



/*A hexadecimal counter is the heart of the
locking algorithm. It will synchronously return
to zero whenever a transition is detected on
the input data stream. It outputs its terminal
count, which can be used to enable the bit
counter. Terminal count is set high when the
counter rolls over after reaching 15 or when
a data transition is detected and the count has
passed its midpoint.*/

module hex_cnt(input CLK, RST, SYNC, 
  output logic TC);
  logic [3:0] COUNT;
  always_ff @(posedge CLK, negedge RST)
    if (~RST) COUNT <= 0;
    else 
      if (SYNC) COUNT <= 0;
      else COUNT <= COUNT + 1;
        
   always_comb TC = (COUNT == 15 || (SYNC && COUNT > 7));
   
endmodule
FIGURE 12.31 Locking counter


/*The frame counter determines where the input data stream
is in the frame in terms of time slots. Each receiving device 
is enabled for just one time slot. The count output is compared
to the slot input to determine if the data being sent over the
common bus are for this device or some others.*/ 

module frame_cnt(input CLK, RST, CLR, EN, output reg [8:0] FRAME);
  always_ff @(posedge CLK, negedge RST)
    if (!RST) FRAME <= 0;
    else
      if (FRAME == 512 && EN) FRAME <= 0;
      else if (CLR) FRAME <= 0;
      else if (EN) FRAME <= FRAME + 1;
      else FRAME <= FRAME;
endmodule
FIGURE 12.32 Frame counter


/*The detector finds transitions on the synchronized data
stream and sends out a synchronization pulse when the start
of a new bit is found. It operates at 16 times the nominal
data rate.*/

module detect(input CLK, DIN, output reg SYNC_DATA, SYNC_PULSE);
  always_ff @(posedge CLK) SYNC_DATA <= DIN;
  always_comb SYNC_PULSE = SYNC_DATA != DIN;
endmodule
FIGURE 12.33 Transition detector


/*The data register is a 10 bit shift register that accepts the
serial data packet input and outputs a parallel byte and a validation
signal when the packet is determined to be correctly formatted.*/

module data_reg(input CLK, RST, DATA_IN, EN, FRAME_EN, input [3:0] BITCNT, 
  input [8:0] FRAME, SLOT, 
  output logic [7:0] PACKET, output logic VALID);
  reg [9:0] SHIFTREG;
  always_ff @(posedge CLK)
    if (BITCNT == 10)
      SHIFTREG <= SHIFTREG;
    else if (FRAME == SLOT && EN) begin
      SHIFTREG[0] <= DATA_IN;
      SHIFTREG[9:1] <= SHIFTREG[8:0];
    end
    else SHIFTREG <= SHIFTREG;
      
    always_comb PACKET = SHIFTREG[9:2];
    always_ff @(posedge CLK, negedge RST)
      if (!RST) VALID = 1'b0;
      else
        if (FRAME == SLOT && EN && BITCNT != 10)
          VALID <= 1'b0;
        else if (SHIFTREG[1:0] == 2'b11 && (FRAME == SLOT) && EN)
          VALID <= 1'b1;
        else VALID <= VALID;
endmodule
FIGURE 12.34 Shift register and data validator




/*This counter tracks the bit position within a slot.
Because there may be an indeterminate number of logic one
bits received following the required two stop bits, the
counter holds its terminal count of 10 until the start of
a new slot causes it to return to zero.*/

module bit_cnt(input CLK, RST, EN, CLR, SYNC_DATA, 
  output logic [3:0] CNT, output logic TC);
  always_ff @(posedge CLK, negedge RST)
    if (!RST) CNT <= 0;
    else
      if (EN && CNT == 10 && !SYNC_DATA) CNT <= 0;
      else if (EN && CLR) CNT <= 0;
      else if (EN && CNT < 10) CNT <= CNT + 1;
      else CNT <= CNT;
  always_comb TC = CNT == 10;
endmodule
FIGURE 12.35 Bit counter


module tb_dmx;
  timeunit 1us;
  timeprecision 1ns;
  //Both periods should work for 3.92 <= period <= 4.08
  //They do not need to be the same. The capture range
  //must be +/- 2% for each.
  parameter PERIOD = 4.0;
  //Real is needed to prevent integer rounding when 16x clock is created
  parameter CLKPERIOD = 4.0;
  //STOPBITS can be any integer >= 2
  parameter STOPBITS = 3;
  //SPACE can be any integer >= 23
  parameter SPACE = 23;
  //MARK can be any integer >= 2
  parameter MARK = 4;
  reg CLK, RST, DATA_IN;
  reg [8:0] SLOT = 1; //UUT will be in slot 1.
  wire [7:0] PACKET;
  wire VALID;
  
  //The Unit Under Test
  dmx UUT(CLK, RST, DATA_IN, SLOT, PACKET, VALID);
  
  task sendframe;
    DATA_IN = 1'b0; //Start Space for Break
    #(SPACE * PERIOD) DATA_IN = 1'b1; //Start Mark for Break
    #(MARK * PERIOD);
  endtask

  task sendpacket(input [7:0] DATA);
    DATA_IN = 1'b0; //Logic 0 Start Bit
    #(PERIOD); //Send packet MSB first
    for (int I = 7; I >= 0; I--) begin
      DATA_IN = DATA[I];
      #(PERIOD);
    end
    DATA_IN = 1'b1; //stop bits
    #(STOPBITS * PERIOD) ;
  endtask
  FIGURE 12.36 Test fixture part 1


initial begin //16x Clock
    CLK = 1'b1;
    forever #(CLKPERIOD/32.0) CLK = ~CLK;
  end
  
  initial begin
    RST = 1'b1; DATA_IN = 1'b1; //Toggle reset
    #(3 * PERIOD) RST = 1'b0;
    #(3 * PERIOD) RST = 1'b1;
    sendframe;
    sendpacket(0); //Slot 0 data
    sendpacket(8'hAA); //Slot 1 data
    sendframe;
    sendpacket(8'hFF);
    sendpacket(8'h55);
    sendframe;
    sendpacket(8'h0);
    sendpacket(8'hDE);
  end
  
  initial begin
    @(posedge VALID)
    if (PACKET != 8'hAA) $display ("Error: Received %h instead of AA", PACKET);
    @(posedge VALID)
    if (PACKET != 8'h55) $display ("Error: Received %h instead of 55", PACKET);
     @(posedge VALID)
    if (PACKET != 8'hDE) $display ("Error: Received %h instead of DE", PACKET);
  end
[bookmark: _GoBack]endmodule
FIGURE 12.37 Test fixture part 2
