module define1(output reg CLK);
  `define PERIOD 10
  initial begin
    CLK = 1'b0;
    forever #`PERIOD CLK = ~CLK;
  end
endmodule
Figure 4.1: Defining a value


module ifdirectives;
  `define ABC //Define but do not give a value to string ABC
  `ifdef ABC //execute the following code if ABC has been defined
    //if defined code goes here
  `elsif //execute the following code if the previous `ifdef was false
    //alternate code goes here
  `endif //end of conditional execution block
  `ifndef ABC //execute the following code if ABC has not been defined
    //if not defined code goes here
  `endif //needed to terminate `ifndef block
  `undef ABC //ABC is no longer defined
endmodule
Figure 4.2: Define directives


module define2;
  `include "definitions.v"
  `define FAULT
  `ifdef FAULT
    MYDES_FAULTED UUT(A, B, CLK, Q);
  `else
    MYDES UUT(A, B, CLK, Q);
  `endif
endmodule
Figure 4.3: Conditional compilation





module define3(output reg CLK = 1'b1);
  `ifndef PERIOD
    `define PERIOD 10
  `endif
  
    initial forever #`PERIOD CLK = ~CLK;
endmodule
Figure 4.4: Checking if a macro has not already been defined


`define TRUE 1’b1
`define FALSE 1’b0
`define PERIOD 10
Figure 4.5: A definitions file


module use_include(input A, B, output reg C);
  `include "definitions.v"
  //TRUE, FALSE and PERIOD are defined in definitions.v
  reg CLK = `TRUE; //initial assignments are non-synthesizable
  initial forever CLK = #`PERIOD ~CLK;
  always @(posedge CLK)
    if (A == `FALSE) C <= A & B;
    else C <= A | B;
endmodule
Figure 4.6: Including a definitions file


module scalable_counter (CLK, RST, CNT);
  parameter SIZE = 4;
  input CLK, RST, 
  output [SIZE - 1 : 0] CNT);
  reg [SIZE - 1 : 0] CNT);
  
always @(posedge CLK or negedge RST)
    if (!RST) CNT <= 0;
    else CNT <= CNT + 1;
endmodule
Figure 4.7: Parameterized (scalable) counter




module scalable_counter #(parameter SIZE = 4) 
(input CLK, RST, output reg [SIZE - 1 : 0] CNT);
  
  always_ff @(posedge CLK, negedge RST)
    if (!RST) CNT <= 'b0;
    else CNT <= CNT + 1;
endmodule
Figure 4.8: Scalable counter using Verilog 2001 and SystemVerilog enhancements


module scalable_counter #(parameter SIZE = 4, MAX_CNT = 11) 
(input CLK, RST, output reg [SIZE - 1 : 0] CNT);
  
  always_ff @(posedge CLK, negedge RST)
    if (!RST) CNT <= 'b0;
    else
      if (CNT == MAX_CNT) CNT <= 0;
      else CNT <= CNT + 1;
endmodule
Figure 4.9: Scalable, programmable counter


module counters(input CLK, RST, 
 output wire [5:0] CNT1, output wire [7:0] CNT2);
 
 scalable_counter #(.SIZE(6), .MAXCNT(31)) C1(CLK, RST, CNT1);
 scalable_counter #(.SIZE(8), .MAXCNT(255)) C2(CLK, RST, CNT2);
 
endmodule
Figure 4.10: Overriding the default value of parameters


module counters(input CLK, RST, 
 output wire [5:0] CNT1, output wire [7:0] CNT2);
 scalable_counter #(6, 31) C1(CLK, RST, CNT1);
 scalable_counter #(8, 255) C2(CLK, RST, CNT2);
endmodule
Figure 4.11: Overriding default values via value rather than named assignment




module counters(input CLK, RST, 
 output wire [5:0] CNT1, output wire [7:0] CNT2);
 
 scalable_counter #(.SIZE(6), .MAXCNT(31)) C1(CLK, RST, CNT1);
 defparam C2.SIZE = 8;
 scalable_counter C2(CLK, RST, CNT2);
endmodule
Figure 4.12: Changing a parameter with defparam


module scale_reg #(WIDTH = 1) (input CLK, 
input [WIDTH - 1 : 0] D, 
output reg [WIDTH - 1 : 0] Q);
  always @(posedge CLK) Q <= D;
  defparam tb_scale_reg.SIZE = 16;
endmodule

module tb_scale_reg;
  parameter SIZE = 4;
  wire [SIZE - 1 : 0] Q;
  reg [SIZE - 1 : 0] D;
  reg CLK = 1'b1;
  initial forever #1 CLK = ~CLK;
  initial begin
    D = 15;
    #2 D = 123;
  end
  scale_reg #(.WIDTH(SIZE)) SR(CLK, D, Q);
endmodule
Figure 4.13: Risky use of defparam to change a value in a higher-level module




module counters(input CLK, RST, 
 output wire [5:0] CNT1, output wire [7:0] CNT2);
 
 defparam C1.SIZE = 5;
 scalable_counter C1(CLK, RST, CNT1);
 defparam C2.SIZE = 8;
 scalable_counter C2(CLK, RST, CNT2);
 //oops: another override of C2 parameter
 defparam C2.SIZE = 16;
 scalable_counter C3(CLK, RST, CNT3);
endmodule
Figure 4.14: Last one wins: C2 will be set to 16 wide; C3 will be left at its default value


module localp(input CLK, RST, output reg [3:0] STATE);
  localparam IDLE = 4'b0001;
  localparam FETCH = 4'b0010;
  localparam DECODE = 4'b0100;
  localparam EXECUTE = 4'b1000;
  
  always_ff @(posedge CLK, negedge RST)
    if (!RST) STATE <= IDLE;
    else begin
      case (STATE)
        IDLE: STATE <= FETCH;
        FETCH: STATE <= DECODE;
        DECODE: STATE <= EXECUTE;
        EXECUTE: STATE <= FETCH;
      endcase
    end
endmodule
Figure 4.15: Localparam example




module etypes;
  enum {FETCH = 1, DECODE, EXEC} STATE;
  initial begin
    STATE = FETCH;
    #1 STATE = DECODE;
    #1 STATE = 3; //This will produce an error
  end
endmodule
Figure 4.16: Illegal assignment to enumerated type variable STATE


module enumsm(CLK, RST, STATE);
  input CLK, RST;
  typedef enum reg [3:0] {IDLE = 4'b0001, FETCH = 4'b0010, 
  DECODE = 4'b0100, EXECUTE = 4'b1000} STATES;
  output STATES STATE;

  always_ff @(posedge CLK, negedge RST)
    if (!RST) STATE <= IDLE;
    else begin
      case (STATE)
        IDLE: STATE <= FETCH;
        FETCH: STATE <= DECODE;
        DECODE: STATE <= EXECUTE;
        EXECUTE: STATE <= FETCH;
      endcase
    end
endmodule
Figure 4.17: Sized enumerated type state machine



module etypes;
  typedef enum {FETCH = 1, DECODE, EXEC} STATE;
  STATE STATES;
  initial begin
    STATES = FETCH;
    #1 STATES = DECODE;
    #1 STATES = STATE'(3); //Casting to enumerated type
  end
endmodule
Figure 4.18: Casting an integer to an enumerated type
module methods(input CLK, RST);
  typedef enum {FETCH, DECODE, EXEC} STATE;
  STATE STATES;
  always_ff @(posedge CLK, negedge RST)
    if (!RST) STATES <= FETCH;
    else STATES <= STATES.next
endmodule
Figure 4.19: Using a method to step through enumerated types


module methods(input CLK, RST);
  typedef enum {FETCH, DECODE, EXEC} STATE;
  STATE STATES;
  always_ff @(posedge CLK, negedge RST)
    if (!RST) STATES <= FETCH;
    /*Bad idea: will result in STATES incrementing to 3 
    instead of wrapping back to FETCH*/
    else STATES <= STATES’{STATES + 1};
endmodule
Figure 4.20: Defective attempt to step through an enumerated type


module methods(input CLK, RST, input [1:0] VAL);
    typedef enum {FETCH, DECODE, EXEC} STATE;
  STATE STATES;
  always_ff @(posedge CLK, negedge RST)
    if (!RST) STATES <= FETCH;
    else begin
        //increment by two
      if (VAL == 0) STATES <= STATES.next(2);
        //decrement by three
      else if (VAL == 1) STATES <= STATES.prev(3);
        //set STATES to the first value, FETCH
      else if (VAL == 2) STATES <= STATES.first;
        //set STATES to the last value, EXEC
      else STATES <= STATES.last;
    end
endmodule
Figure 4.21: Using enumerated type methods




module COUNT(CLK, RST,CNT);
  input CLK, RST;
  typedef enum {ZERO, ONE, TWO, THREE} NUMBERS;
  output NUMBERS CNT;

  always_ff @(posedge CLK, negedge RST)
    if (!RST) CNT <= ZERO;
  else CNT <= CNT.next;
endmodule
Figure 4.22: Counter using enumerated type


module DECODE(CNT, ONEHOT);
  typedef enum {ZERO, ONE, TWO, THREE} NUMBERS;
  input NUMBERS CNT;
  output logic [3:0] ONEHOT;
  always_comb
    case (CNT)
      ZERO: ONEHOT = 4'b0001;
      ONE: ONEHOT = 4'b0010;
      TWO: ONEHOT = 4'b0100;
      THREE: ONEHOT = 4'b1000;
      default: ONEHOT = 4'bx;
    endcase
endmodule
Figure 4.23: Enumerated type is used an input


module TOP(input CLK, RST, output logic [3:0] ONEHOT);
   typedef enum {ZERO, ONE, TWO, THREE} NUMBERS;
   NUMBERS CNT;
   COUNT  C(CLK, RST, CNT);
   DECODE D(CNT, ONEHOT);
endmodule
Figure 4.24: Enumerated types must be specified between the modules, too





module tb_top;
  bit CLK, RST;
  logic [3:0] ONEHOT;
  TOP  UUT(CLK, RST, ONEHOT);
   initial forever #1 CLK = ~CLK;
   initial begin
     RST = 1'b1;
     #2 RST = 1'b0;
     #3 RST = 1'b1;
   end
endmodule
Figure 4.25: Test fixture for enumerated type circuit design


package globe;
  `define true 1’b1;
  `define false 1’b0
  `define period 10
endpackage
Figure 4.27: Global definitions in a package


module useglobe(input A, output logic B);
  import globe::*; //import everything in package globe
  always_comb
    if (A == `true) B = `false;
    else B = `true;
endmodule
Figure 4.28: Importing all elements of a package




package globe;
  parameter true = 1'b1; 
  parameter false = 1'b0;
  parameter period = 10;
endpackage

module useglobe(input A, output logic B);
  import globe::true;
  import glove::false;
  always_comb
    if (A == true) B = false;
    else B = true;
endmodule
Figure 4.29: Importing individual elements of a package






package types;
   typedef enum {ZERO, ONE, TWO, THREE}NUMBERS;
endpackage
 
module COUNT(CLK, RST, CNT);
  input CLK, RST;
  import types::*;
  output NUMBERS CNT;
  
  always_ff @(posedge CLK, negedge RST)
    if (!RST) CNT <= ZERO;
    else CNT <= CNT.next;
endmodule
  
module DECODE(CNT, ONEHOT);
  import types::*;
  input NUMBERS CNT;
  output logic [3:0] ONEHOT;
  always_comb
    case (CNT)
      ZERO: ONEHOT = 4'b0001;
      ONE: ONEHOT = 4'b0010;
      TWO: ONEHOT = 4'b0100;
      THREE: ONEHOT = 4'b1000;
      default: ONEHOT = 4'bx;
    endcase
endmodule
  
module TOP(input CLK, RST, output logic [3:0] ONEHOT);
   import types::*;
   NUMBERS CNT;
   COUNT  C(CLK, RST, CNT);
   DECODE D(CNT, ONEHOT);
endmodule
Figure 4.30: Hierarchical design with a package





[bookmark: _GoBack]module ALLONES;
  parameter SIZE = 64;
  reg [SIZE - 1 : 0] A, B, C, D;
  initial begin
    //A, B C and D will all be set to 64 bits of logic one
    A = -1;
    B = ~0;
    C = {SIZE{1'b1}};
    D = '1;
  end
endmodule
Figure 4.31: Setting all bits of a variable to logic one
