module TOPDOWN(input CLK, RST, INT, inout [31:0] BIBUS,
  output RD, AD);
  
  //instances of functional blocks go here
  
endmodule
Figure 3.1: Top level of a machine that has not been fully defined


module BLOCKS;
  
  reg CLK, A, B;
  
  initial begin
    //non-synthesizable statements
  end
  
  always_ff @(posedge CLK) begin
    //synthesizable SystemVerilog sequential statements
  end
  
  always_comb begin
    //synthesizable SystemVerilog combinational statements
  end
  
  always_latch begin
    //synthesizable SystemVerilog behavioral statements with
    //latched outputs
  end
  
  always @(*) begin
    //Verilog combinational statements
  end
  
  always @(A or B) begin
    //Verilog 95 style combinational statements
  end
  
endmodule
Figure 3.2: Verilog and SystemVerilog functional block statements






module namedblock;
  
  reg [3:0] COUNT; //module variable COUNT
  
  always_comb begin: BLOCK1
    reg [7:0] COUNT; //BLOCK1 variable COUNT
  end : BLOCK1
  
  always_comb begin
    //This block will use the module variable COUNT
  end
  
endmodule
Figure 3.3: Named block with a local variable


module BADLIST(input A, B, output reg Y, Z);
  always @(A) Y = A ^ B; //XOR, defective sensitivity list
  always @(A, B) Z = A ^ B; //XOR, complete sensitivity list
endmodule

Time 0: A = 0, B = 0, Y = 0, Z = 0
Time 1: A = 0, B = 1, Y = 0, Z = 1
Time 2: A = 1, B = 0, Y = 1, Z = 1
Time 3: A = 1, B = 1, Y = 1, Z = 0
Figure 3.4: Defective sensitivity list and memory effect in simulation.

module shortcicuit(input CLK, RST, D, output reg Q);
/*Design error: this would appear to work in simulation
but would infer two flipflops shorted together in hardware.*/
  always @(negedge RST) Q <= 1'b0;
  always @(posedge CLK) Q <= D;
endmodule
Figure 3.5: Two flipflops with outputs shorted together




Table 3.1: Verilog net types
	Net Type
	Function

	wire
	Normal interconnects between instances

	tri
	Exactly the same as wire

	wand
	Wired AND. Models open drain/open collector devices

	triand
	Exactly the same as wand

	wor
	Wired OR. Used in now-obsolete emitter-coupled logic

	trior
	Exactly the same as wor

	trireg
	Models nets with capacitive storage.  Holds last driven value

	tri0
	Nets that pull down when not driven

	tri1
	Nets that pull up when not driven






[bookmark: _GoBack]reg A; //Single bit unsigned register variable
reg [15:0] B, C; //Two 16-bit unsigned vectors, 15 is MSB
reg [0:31] D; //One 32-bit unsigned vector, 0 is MSB
reg signed [15:0] E, F; //Two 16-bit signed vectors
Figure 3.6: Sample register declarations




module SIGNED_ASSIGN (A, X, Y);
  input signed [7:0] A;
  output reg signed [15:0] X, Y;

  always @(A) begin
    X = A; //A will be sign extended to 16 bits
    Y = A[7:0]; //No sign extension
  end
Figure 3.7: Bit selects are always unsigned



 module vectorfill;
  reg [31:0] W;
  reg [15:0] X;
  reg [7:0] Y;
  reg Z;
  
  always_comb begin
    W = '0; //W will be 32 bits of zeros
    X = '1; //X will be 16 bits of ones
    Y = 'X; //Y will be eight bits of unknown
    Z = 'Z; //Z will be one bit set to high impedance
  end
endmodule
Figure 3.8: Filling a vector with all 0, 1, X or Z values


module array_decs;
  
  //a one kilobyte register file
  reg [7:0] MEM [0:1023];
  
  //an array of 100 16-bit integers
  shortint NUMBERS [0:99];
  
  //an array of 32 one-bit regs
  reg BIT_ARRAY [0:31];
  
  //declare only the size, not the range
  logic [15:0] SVMEM [512]; //elements will be 0 to 511
  
endmodule
Figure 3.9: Array declarations




module array_assign(input CLK, RST, 
  input [3:0] DATA1, DATA2, 
  output reg [3:0] MEM [4]);
  
  always_ff @(posedge CLK, negedge RST)
	if (!RST) MEM <= '{default:0};
	else MEM <= '{4'hA, DATA1, DATA2, 4'b1010};
endmodule
Figure 3.10: Assigning to all elements of an array with single statements


module array_funcs;
  reg [3:0] MEM[4] [5] [6];
  int A, B, C, D, E, F, G;
  
  assign A = $left(MEM[2][2][2]);
  assign B = $right(MEM[2][2][2]);
  assign C = $low(MEM[2][2][2]);
  assign D = $high(MEM[2][2][2]);
  assign E = $increment(MEM[2][2][2]);
  assign F = $size(MEM[2][2][2]);
  assign G = $dimensions(MEM);
  
  initial $display("A = %d, B = %d, C = %d, D = %d, E = %d, F = %d, G = %d",
  A, B, C, D, E, F, G);
endmodule


/* Simulaton Results:
# A = 3, B = 0, C = 0, D = 3, E = 1, F = 4, G = 4*/
Figure 3.11: Array Functions




/*Array with output connected to a bidirectional bus.*/

module MEMCELL(input OE, WS, input [7:0] ADDR,
  inout wire [15:0] DATA);
  
  /*One 16 bit reg variable(DATA_REG), one 16x256 bit array (MEMORY).
  Only MEMORY will infer flipflops. DATA_REG will be a combinational
  decode of the array, but it is still a Verilog reg variable.*/
  reg [15:0] DATA_REG, MEMORY[0 : 255];
  
  //Registers cannot be connected to an inout port
  assign DATA = OE ? DATA_REG : 16'bz;
  
  //Select one 16 bit element of the memory array
  always_comb DATA_REG = MEMORY[ADDR];
  
  //Write to memory on rising edge of write strobe (WS)
  always_ff @(posedge WS) MEMORY[ADDR] <= DATA;
endmodule
Figure 3.12: Memory cell connected to bidirectional bus



module unn;
  
  union packed {
    reg [7:0] A; //A, B, and C all reference the same space
    bit [7:0] B;
    byte C;
    } AB;
  
  initial begin
    $monitor("%d: A = %h, B = %h, C = %h", $time, AB.A, AB.B, AB.C);
    AB.A = 8'hAA; //B and C are never directly assigned any values
    #1 AB.A = 8'h55;
  end
  
/*Simulation Results:
#0: A = aa, B = aa, C = AA
#1: A = 55, B = 55, C = 55*/
endmodule
Figure 3.13: The union will reference the same space for all three names



module unn;
  
  union packed {
    reg [7:0] A; //A, B, and C all reference the same space
    bit [7:0] B;
    byte C;
    shortint D; //Error: D is a 16-bit value
    } AB;
Figure 3.14: All fields of a packed union must be the same size



module unn2(input CLK, input [7:0] A, output wire [7:0] Z);
  
  union packed {
    reg [7:0] X;
    bit [7:0] Y;
  } XY;
  
  assign Z = XY.Y;
  always_ff @(posedge CLK) XY.X <= A;
    
endmodule
Figure 3.15: Synthesis of module unn2 would yield eight flipflops



module strs2;
  
  typedef struct {bit WS, OE;} CON;
    
  struct {bit [15:0] ADDR, DATA;
    CON CBUS;} BUS;
    
  reg [15:0] MEM [0:1023];
  
  always_ff @(posedge BUS.CBUS.WS) MEM[BUS.ADDR] <= BUS.DATA;
   
endmodule
Figure 3.16: A hierarchical system of structures




module strs3(IBUS, DATA);
  typedef struct {bit [15:0] ADDR;
  bit WS, OE;} BUS;
  input BUS IBUS;
  inout [15:0] DATA;
  
  reg [15:0] MEM [0:1023], DATA_REG;
      
  assign DATA = IBUS.OE ? DATA_REG : 16'hz;
  always_comb DATA_REG = MEM[IBUS.ADDR];
  always_ff @(posedge IBUS.WS) MEM[IBUS.ADDR] <= DATA;
    

endmodule
Figure 3.17: Using a structure in a port list



module strs4;
  
  typedef struct packed {bit WS, OE;} CON;
    
  struct packed {bit [15:0] ADDR, DATA;
    CON CBUS;} BUS;
      
  assign BUS = {16'h4, 16'haa, 1'b0, 1'b1};

endmodule
Figure 3.18: Referencing all fields of a structure


module blocking(input [3:0] A, B, C, D, 
  output reg [4:0] X, Y, output reg [5:0] Z);
  
  always_comb begin
    X = A + B;
    Y = C + D;
    Z = X + Y;
  end
    
endmodule
Figure 3.19: The blocking assignment operator




module sr2(input CLOCK, DATAIN, output reg FF2);
  reg FF1;
  always_ff@(posedge CLOCK) begin
    FF1 = DATAIN;
    FF2 = FF1;
  end
endmodule
Figure 3.20: Faulty use of the blocking assignment operator in a sequential block


module sr2(input CLOCK, DATAIN, output reg FF2);
  reg FF1;
  always_ff@(posedge CLOCK) begin
    FF1 <= DATAIN;
    FF2 <= FF1;
  end
endmodule
Figure 3.23: Code for a shift register using the non-blocking assignment operator


module CS1(input A, B, C, CLOCK, output reg Z);
  reg T1, T2;
  always_comb begin
    T1 = A ^ B;
    T2 = C & T1;
  end
  
  always_ff@(posedge CLOCK)
    Z <= T2;
endmodule
Figure 3.25: Separate combinational and sequential blocks


module CS2(input A, B, C, CLOCK, output reg Z);
  always_ff @(posedge CLOCK) Z <= C & (A ^ B);
endmodule
Figure 3.26: One sequential block


module CS3(input A, B, C, CLOCK, output reg Z);
  reg T1, T2;
  always_ff@(posedge CLOCK) begin
    T1 = A ^ B;
    T2 = C & T1;
    Z <= T2;
  end    
endmodule
Figure 3.27: Legal mixing of assignment operators in a single functional block
module NIX(A, B, C, Z);
  input A, B, C;
  output reg Z;
  always @(A or B or C)
    if (A) Z = A & B;
    else Z <= A | B;
endmodule
Figure 3.28: Illegal mixing of assignment operators



module equality;
  
  reg [2:0] A, B, C, D;
  integer E, F;
  reg U, V, W, X, Y, Z;
  
  initial begin
    A = 3'b11z;
    B = 3'b01x;
    C = 3'b11z;
    D = 3'b011;
    E = 3;
    F = 7;
   
    //Equality operator, no comparison to Z
    U = (A == C); //U = 1'bx
    //Case equality, does compare to Z
    V = (A === C); //V = 1'b1
    //Equality, no comparison to X, either
    W = (B == D); //W = 1'bx
    //Case equality, resolves X is not Z
    X = (B === D); //X = 1'b0
    //Compares integer and vector values
    Y = (D == E); //Y = 1'b1
    //Smaller vector is zero-extended
    Z = (D == F); //Z = 1'b0
  end
endmodule
Figure 3.29: Equality operator examples





module equality2;
  
  reg [4:0] A, B;
  reg Y, Z;
  
  initial begin
    A = 5'b1010x;
    B = 5'b00101;
    Y = (A != B); //A is not equal to B even without the last bit
    Z = (A !== B); //Logical and case inequality work the same in this case
  end
endmodule
Figure 3.30: Resolving equality/inequality from determinate bits alone


module wildcards(input [2:0] A, B, output logic [2:0] C, D);
  always_comb begin
    C = (A ==? 3'b10?); //C will be true for A = 100 and 101;
    D = (B !=? 3'b?00); //D will be true whenever the two LSBs of B are not 00
  end
endmodule
Figure 3.31: Synthesizable wildcard case equality and inequality operators


module logicals;
  reg [7:0] A, B;
	reg C;
	always_comb begin
	  A = 8'b01010101; //A has some bits set, so it is true
	  B = 8'b10101010; //Same for B
	  if (A && B) C = 1; //Both A and B are true, so C will be set
	  else C = 0; 
	end
endmodule
Figure 3.32: Logical operation, AND of 1 and 1 yielding 1




module logicals2;
  reg [7:0] A;
	reg D, E;
	always_comb begin
	  A = 8'b0000000x;
	  if (A) D = 1;
	  else D = 0;
	  if (!A) E = 1;
	  else E = 0;
	end
endmodule 
Figure 3.33: A is neither true nor false. Both D and E will be 0.


module bitwise;
  reg [4:0] A, X;
  reg [3:0] B, V, W, Z;
  reg [2:0] C, D, Y;
  
  initial begin
    A = 5'b10101;
    B = 4'b010x;
    C = 3'b011;
    D = 3'b11z;
    
    V = ~B; //Inverse of X is X
    W = A | B; //B will be zero-extended by one bit
    X = A & B; //Indeterminate bit will cause indeterminate output
    Y = C ^ D; //High impedance bit will act like unknown
    Z = B ^~ C; //XNOR, zero extend C
    $strobe("V = %b, W = %b, X = %b, Y = %b, Z = %b", V, W, Y, Y, Z);
  end
endmodule

/*Simulation Results
# V = 101x, W = 0101, X = 10x, Y = 10x, Z = 100x
*/
Figure 3.34: Bitwise operators






module reduction;
  reg [7:0] A, B, C;
  reg W, X, Y, Z;
  initial begin
    A = 8'hFF; //All ones
    B = 0; //All zeros
    C = 8'b01100000; //Even number of set bits
    W = &A; //AND reduction: detect max. value
    X = |B; //OR reduction: detect all zeros
    Y = ^C; //XOR reduction: positive parity
    Z = ~^C; //XNOR reduction: negative parity
    $strobe("W = %b, X = %b, Y = %b, Z = %b", W, X, Y, Z);
  end
endmodule

/*Simulation Results:
# W = 1, X = 0, Y = 0, Z = 1
*/
Figure 3.35: Reduction operators



module ARITH(input [7:0] A, B, output reg [8:0] SUM, DIF,
  output reg [15:0] PROD, output reg [7:0] QUO, REM,
  output reg [1023:0] EXP);
  
  always_comb begin
    SUM = A + B; //Addition
    DIF = A - B; //Subtraction
    PROD = A * B; //Multiplication
    QUO = A / B; //Division
    REM = A % B; //Modulus
    EXP = A ** B; //Exponential
  end
  
endmodule
Figure 3.36: Arithmetic operators





module divzero;
  const int PERIOD = 1;
  reg CLK;
  initial begin
    CLK = 1'b0;
    //Integer division: 1 / 2 = 0
    #(PERIOD / 2) CLK = 1'b1;
    #(PERIOD / 2) CLK = 1'b0;
    #(PERIOD / 2) CLK = 1'b1;
    #(PERIOD / 2) CLK = 1'b0;
  end
endmodule
Figure 3.37: Integer division truncates



for (I = 0; I < LIMIT; I++) //post increment
for (I = 0; I < LIMIT; I--) //post decrement
for (I = 0; I < LIMIT; ++I) //pre increment
for (I = 0; I < LIMIT; --I) //pre decrement
Figure 3.38: Auto increment and decrement operators used in loop controls



always @(posedge CLK, negedge RST)
  if (!RST) CNT <= ‘b0’ //non-blocking assignment
  else CNT++; //blocking assignment
Figure 3.39: Illegal mixing of blocking and non-blocking assignments




module relational;
  
  integer A, B;
  reg [2 : 0] C;
  reg [1:0] D;
  reg W, X, Y, Z;
  
  initial begin
    A = 5;
    B = 17;
    C = 3'b11x;
    D = 2'b0z;
    
    W = (A < B); //True: W = 1'b1
    X = (B <= A); //False: X = 1'b0
    Y = (C < B); //Will not resolve: Y = 1'bx;
    Z = (D >= A); //Will not resolve: Z = 1'bx;
    $strobe("W = %b, X = %b, Y = %b, Z = %b", W, X, Y, Z);
  end
endmodule

/*Simulation Results:
# W = 1, X = 0, Y = x, Z = x
*/
Figure 3.40: Relational operator examples


module resolve;
  
  reg [3:0] A, B;
  reg X, Y, Z;
  
  initial begin
    A = 4'b1000;
    B = 4'b000x;
    
    //Relational will return unknown
    X = (A > B); //X = 1'bx
    //Logical equality will resolve A is not equal to B
    Y = (A == B); //Y = 1'b0
    //Case equality will also resolve A is not equal to B
    Z = (A === B); //Z = 1'b0
  end
endmodule
Figure 3.41: Difference in resolution between relational and equality operators



always_ff @(posedge CLK)
  for (int I = 0; I <= MAX; I++) //<= is a relational operator here
    if (ENABLE) OUT <= OP1 + OP2; //but it is the non-blocking assignment operator here
    else OUT <= OUT;
Figure 3.42: <= can mean a relational or an assignment operator



module shift;
  reg signed [7:0] A, W, X, Y, Z;
  
  initial begin
    A = 8'hFF; //Initialize to 11111111
    W = A >> 2; //Shift right by 2 bits (divide by 4)
    X = A << 2; //Shift left by 2 bits (multiply by 4)
    Y = A >>> 2; //Arithmetic shift right, sign extend
    Z = A <<< 2; //Arithmetic shift left
    $strobe("A = %b, W = %b, X = %b, Y = %b, Z = %b", A, W, X, Y, Z);
  end
endmodule
Figure 3.43: Shift operations



module barrel(A, SHIFT, OUT);
  parameter DIST = 3;
  parameter WIDTH = 2**DIST;
  input [WIDTH - 1 : 0] A;
  input [DIST - 1 : 0] SHIFT;
  output reg [WIDTH - 1 : 0] OUT;
  
  always @(A, SHIFT) OUT = {A, A} >> SHIFT;
endmodule
Figure 3.44: Barrel shifter using a shift operator





module streaming;
  logic [15:0] A, B, C;
  logic [7:0] D, E;
  initial
    A = 16'h1234;
  always_comb begin
    B = {<<{A}};
    {>>{C}} = A;
    D = {<<{ A[7:0]}};
    {>>{E}} = A;
    $strobe("A = %h, B = %h, C = %h, D = %h, E = %h", A, B, C, D, E);
  end
endmodule

/*Simulation Results:
# A = 1234, B = 2c48, C = 1234, D = 2c, E = 12
*/
Figure 3.45: Streaming operation



module adder(A, B, CIN, SUM, COUT);
  input A, B, CIN;
  output reg SUM, COUT;
  
  always @(A or B or CIN) {COUT, SUM} = A + B + CIN;
endmodule
Figure 3.46: Adder using concatenation


module replication;
  reg [3:0] A; 
  reg [7:0] X, Y, Z;
  
  initial begin
    A = 4'b1010;
    
    X = {2{A}}; //make two copies of A concatenated together
    Y = {{0{A[3]}}, A}; //zero replication constant
    Z = {{4{A[3]}}, A}; //Extend A's MSB four more bits  
  end
  
endmodule
Figure 3.47: Replication examples




module conditional(input A, B, SEL, EN, 
output reg OUT);
  reg C;
  
  always_comb begin
    C = SEL ? A : B; //creat a multiplexor
    OUT = EN ? C : 1'bz; //create a Tri-state driver
  end
endmodule
Figure 3.48: Conditional operators creating a multiplexor and a Tri-state driver



module conditional2;
  reg [2:0] A, B, C;
  reg SEL;
  
  always_comb C = SEL ? A : B;
  
  initial begin
    #0 A = 3'b010; B = 3'b101; SEL = 1'b0;
    #1 SEL = 1'b1;
    #1 SEL = 1'bx;
    #1 B = 3'b011;
  end

initial $monitor($time, "A = %b, B = %b, C = %b, SEL = %b", A, B, C, SEL);

endmodule

/*Simulation Results:
# Time 0: A = 010, B = 101, C = 101, SEL = 0
# Time 1: A = 010, B = 101, C = 010, SEL = 1
# Time 2: A = 010, B = 101, C = xxx, SEL = x
# Time 3: A = 010, B = 011, C = 01x, SEL = x
*/
Figure 3.49: Multi-bit multiplexor resolving common bits


module combined;
  reg signed [3:0] A, B, C, X, Y, Z;
  
  initial begin
    A = 4'b1010;
    B = 4'b0001;
    C = 4'd2;
    X = 4'h5;
    Y = 4'b1;
    Z = 4'b1111;
    X += A; //Add X + A, assign to X
    Y <<= B; //Shift Y left by B bits
    Z >>>= C; //Arithmetic shift right Z by C bits
  end
endmodule
Figure 3.50: Combined assignment operators



