`timescale 1 ns / 1 ns
module race (OUT1, OUT2, CLK, RST);
output OUT1, OUT2;
input CLK, RST;
reg OUT1, OUT2;
    
always @(posedge CLK or negedge RST)
  if (!RST) OUT1 = 0;
  else OUT1 = OUT2;
    
always @(posedge CLK or negedge RST)
  if (!RST) OUT2 = 1;
  else OUT2 = OUT1;

endmodule
FIGURE 8.3 Race condition in behavioral code


`timescale 1 ns / 1 ns
module race (OUT1, OUT2, CLK, RST);
output OUT1, OUT2;
input CLK, RST;
reg OUT1, OUT2;
    
always @(posedge CLK or negedge RST)
  if (!RST) OUT2 = 1;
  else OUT2 = OUT1;
    
always @(posedge CLK or negedge RST)
  if (!RST) OUT1 = 0;
  else OUT1 = OUT2;

endmodule
FIGURE 8.4 Same race, different outcome




module race2;    
  reg a2;
    
initial begin
    a2 = 1'b0;
    #10 a2 = 1'b1;
  end

  initial
    #10 if (a2) $display("This may not print");

endmodule
FIGURE 8.5 Results will depend on queuing order



module race2;    
  reg a2;
  
  initial
    #10 #0 if (a2)
    $display("This one is different. It will print");

  initial
    #10 if (a2) $strobe("This may not print");

  initial begin
    a2 = 1'b0;
    #10 a2 = 1'b1;
  end

endmodule
FIGURE 8. 6 Addition of a zero delay will put the evaluation into a different queue




module race3(input CLK, RST, output logic [3:0] CNT, 
  output logic MATCH);
  logic CLK2;
  const int TC = 15;
  
  /*When this block is last, MATCH is asserted when CNT = 15.
  When it is first, MATCH is asserted when CNT = 0.
  If all sequential assignments are made with non-blocking
  assignment operators, the order of the blocks does not
  matter and MATCH is always asserted when CNT is 15. That
  is also the case when all blocks use the master clock
  rather than having one use CLK2.*/
  always_ff @(posedge CLK, negedge RST) begin
    //Derived clock, 1/2 frequency of master clock
    if (!RST) CLK2 = 1'b0;
    else CLK2 = ~CLK2;
  end
  
  //This block uses the derived clock
  always_ff @(posedge CLK2, negedge RST) begin
    if (!RST)
      CNT = '0;
    else
      if (CNT == TC) CNT = '0;
      else CNT = CNT + 1;
  end
  
  //This block uses the master clock
  always_ff @ (posedge CLK) begin
  //Match signal is "and"ed with CLK2 so it
  //will be active for only one master clock cycle
    if (CNT == TC && CLK2) MATCH = 1'b1;
    else MATCH = 1'b0;
  end 
  
endmodule
FIGURE 8.7 Using a logic signal as a clock




module race4(input CLK, RST, output logic [3:0] CNT,
  output logic MATCH);
  logic CLK2;
  localparam TC = 15;
  
  //Enable every other clock cycle
  always_ff @(posedge CLK, negedge RST)
    if (!RST) CLK2 <= 1'b0;
    else CLK2 <= ~CLK2;
      
  always_ff @(posedge CLK, negedge RST)
    if (!RST) CNT <= 0;
    else
      if (CLK2 && CNT == TC) CNT <= 0;
      else if (CLK2) CNT <= CNT + 1;
      else CNT <= CNT;
        
  always_ff @(posedge CLK)
    //Match signal is "and"ed with CLK2 so it
    //will be active for only one master clock cycle
    if (CNT == TC && !CLK2) MATCH <= 1'b1;
    else MATCH <= 1'b0;
      
endmodule
[bookmark: _GoBack]FIGURE 8.8 Synchronous design with nonblocking assignment operators
